*******		REPLICATION FILES
*******		Climate Variability and Irregular Migration to the European Union 
*******		Global Environmental Change 
*******		Fabien Cottier and Idean Salehyan
*******		Replication: Sensitivity analysis (S11) splitting countries by geographical location (Tables A.35-36, Figure A.26)
*******		This version: April 2021

* note: stata packages required for running models
* - coefplot
* - crossfold
* - plottig


clear all

set more off
set scheme plottig
cd "path/to/replication/directory"

* load data
use "data_quarter.dta", clear 

* load stata risk scripts
qui do scripts/script_rr_spei_location_102020

* duplicates data
duplicates list cowc year quarter
duplicates tag cowc year quarter, gen(_dupl)
duplicates drop cowc year quarter, force

* set time series
sort cowc year quarter
gen quarter_int = real(regexs(1)) if regexm(quarter,"([0-9]+)")
gen quarterS=(year-2005)*4+quarter_int
order cowc nationality year quarter quarter_int quarterS
tsset cowc quarterS
 
* gen additional variables
encode continent, gen(contN)
recode contN (5=.) // Africa is continent 1


* generate lag quartely migration variables
by cowc: gen nmigrq_exbalk_1tl = nmigrq_exbalk[_n-1]
by cowc: gen nmigrq_exbalk_2tl = nmigrq_exbalk[_n-2]
by cowc: gen nmigrq_exbalk_3tl = nmigrq_exbalk[_n-3]
by cowc: gen nmigrq_exbalk_4tl = nmigrq_exbalk[_n-4]
by cowc: gen nmigrq_exbalk_ln_1tl = nmigrq_exbalk_ln[_n-1]
by cowc: gen nmigrq_exbalk_ln_2tl = nmigrq_exbalk_ln[_n-2]
by cowc: gen nmigrq_exbalk_ln_3tl = nmigrq_exbalk_ln[_n-3]
by cowc: gen nmigrq_exbalk_ln_4tl = nmigrq_exbalk_ln[_n-4]


* generate dummies variable for extreme weather (cut-off 10 percentile / 90 percentile)
qui xtreg nmigrq_exbalk_ln nmigrq_exbalk_ln_* wmean_speiy /// 
 i.year i.quarter_int if migr100_exbalk==1, ///
 fe vce(cluster cowc)

centile (wmean_speiy) if e(sample), centile (10 90)
gen spei_drought = 0 if wmean_speiy!=.
recode spei_drought (0=1) if wmean_speiy <= r(c_1) 
centile (wmean_speiy) if e(sample), centile (10 90)
gen spei_hrain = 0 if wmean_speiy!=.
recode spei_hrain (0=1) if wmean_speiy >=  r(c_2) 


*******		  Macros

global SPEI_Y0 wmean_speiy

global SPEI_SQ_Y0 c.wmean_speiy##c.wmean_speiy
 
global SPEI_Y0Y2 L(0 4 8).wmean_speiy

global SPEI_SQ_Y0Y2 c.wmean_speiy##c.wmean_speiy cL4.wmean_speiy##cL4.wmean_speiy /// 
  cL8.wmean_speiy##cL8.wmean_speiy




*******		 Table A.35


* NULL Model // agrarian countries
qui xtreg nmigrq_exbalk_ln nmigrq_exbalk_ln_* $SPEI_Y0 ///
	i.year i.quarter_int if migr100_exbalk==1 & contN==1, ///
	fe vce(cluster cowc)
qui xtreg nmigrq_exbalk_ln nmigrq_exbalk_ln_* ///
	i.year i.quarter_int if e(sample), ///
	fe vce(cluster cowc)
est sto tabA35_m0H
* cross validation / RMSE
local reg_call=e(cmdline)
qui crossfold `reg_call'
mata : st_numscalar("CVrmse", sqrt(mean((st_matrix("r(est)")):^2)))
di as text   "Avg rmse null model     : " CVrmse
* list of countries
est resto tabA35_m0H
unique cowc if e(sample)
tabulate nationality contN if e(sample)

* NULL Model // non-agrarian countries
qui xtreg nmigrq_exbalk_ln nmigrq_exbalk_ln_* $SPEI_Y0 ///
	i.year i.quarter_int if migr100_exbalk==1 & contN!=1, ///
	fe vce(cluster cowc)
qui xtreg nmigrq_exbalk_ln nmigrq_exbalk_ln_* ///
	i.year i.quarter_int if e(sample), ///
	fe vce(cluster cowc)
est sto tabA35_m0L
* cross validation / RMSE
local reg_call=e(cmdline)
qui crossfold `reg_call'
mata : st_numscalar("CVrmse", sqrt(mean((st_matrix("r(est)")):^2)))
di as text   "Avg rmse null model     : " CVrmse
global CVrmsenull_LA=round(CVrmse,0.001)
* list of countries
est resto tabA35_m0L
unique cowc if e(sample)
tabulate nationality contN if e(sample)


* Model 5 // agrarian countries
xtreg nmigrq_exbalk_ln nmigrq_exbalk_ln_* $SPEI_Y0 ///
  i.year i.quarter_int if migr100_exbalk==1 & contN==1, ///
  fe vce(cluster cowc)
est sto tabA35_m5
* F test spei parameters & aic
test wmean_speiy
estat ic
* cross validation / RMSE
local reg_call=e(cmdline)
qui crossfold `reg_call'
mata : st_numscalar("CVrmse", sqrt(mean((st_matrix("r(est)")):^2)))
di as text   "Avg rmse     : " CVrmse

* Model 6 // non-agrarian countries
xtreg nmigrq_exbalk_ln nmigrq_exbalk_ln_* $SPEI_Y0 ///
  i.year i.quarter_int  if migr100_exbalk==1 & contN!=1, ///
  fe vce(cluster cowc)
est sto tabA35_m6
* F test spei parameters & aic
test wmean_speiy
estat ic
* cross validation / RMSE
local reg_call=e(cmdline)
qui crossfold `reg_call'
mata : st_numscalar("CVrmse", sqrt(mean((st_matrix("r(est)")):^2)))
di as text   "Avg rmse     : " CVrmse


* Model 7 // agrarian countries
xtreg nmigrq_exbalk_ln nmigrq_exbalk_ln_* $SPEI_SQ_Y0 ///
  i.year i.quarter_int  if migr100_exbalk==1 & contN==1, ///
  fe vce(cluster cowc)
est sto tabA35_m7
* F test spei parameters & aic
test wmean_speiy c.wmean_speiy#c.wmean_speiy
estat ic
* cross validation / RMSE
local reg_call=e(cmdline)
qui crossfold `reg_call'
mata : st_numscalar("CVrmse", sqrt(mean((st_matrix("r(est)")):^2)))
di as text   "Avg rmse     : " CVrmse

* Model 8 // non-agrarian countries
xtreg nmigrq_exbalk_ln nmigrq_exbalk_ln_* $SPEI_SQ_Y0 ///
  i.year i.quarter_int  if migr100_exbalk==1 & contN!=1, ///
  fe vce(cluster cowc)
est sto tabA35_m8
* F test spei parameters & aic
test wmean_speiy c.wmean_speiy#c.wmean_speiy
estat ic
* cross validation / RMSE
local reg_call=e(cmdline)
qui crossfold `reg_call'
mata : st_numscalar("CVrmse", sqrt(mean((st_matrix("r(est)")):^2)))
di as text   "Avg rmse     : " CVrmse


* Model 9 // agrarian countries
xtreg nmigrq_exbalk_ln nmigrq_exbalk_ln_* $SPEI_Y0Y2 ///
  i.year i.quarter_int  if migr100_exbalk==1 & contN==1, ///
  fe vce(cluster cowc)
est sto tabA35_m9
unique cowc if e(sample)
* F test spei parameters & aic
test L0.wmean_speiy L4.wmean_speiy L8.wmean_speiy
estat ic
* cross validation / RMSE
local reg_call=e(cmdline)
qui crossfold `reg_call'
mata : st_numscalar("CVrmse", sqrt(mean((st_matrix("r(est)")):^2)))
di as text   "Avg rmse     : " CVrmse

* Model 10 // non-agrarian countries
xtreg nmigrq_exbalk_ln nmigrq_exbalk_ln_* $SPEI_Y0Y2 ///
  i.year i.quarter_int  if migr100_exbalk==1 & contN!=1, ///
  fe vce(cluster cowc)
est sto tabA35_m10
unique cowc if e(sample)
* F test spei parameters & aic
test L0.wmean_speiy L4.wmean_speiy L8.wmean_speiy
estat ic
* cross validation / RMSE
local reg_call=e(cmdline)
qui crossfold `reg_call'
mata : st_numscalar("CVrmse", sqrt(mean((st_matrix("r(est)")):^2)))
di as text   "Avg rmse     : " CVrmse


* Model 11 // agrarian countriese
xtreg nmigrq_exbalk_ln nmigrq_exbalk_ln_* $SPEI_SQ_Y0Y2 ///
  i.year i.quarter_int  if migr100_exbalk==1 & contN==1, ///
  fe vce(cluster cowc)
est sto tabA35_m11
* F test spei parameters & aic
test L0.wmean_speiy L4.wmean_speiy L8.wmean_speiy ///
 cL0.wmean_speiy#cL0.wmean_speiy cL4.wmean_speiy#cL4.wmean_speiy cL8.wmean_speiy#cL8.wmean_speiy
estat ic
* cross validation / RMSE
local reg_call=e(cmdline)
qui crossfold `reg_call'
mata : st_numscalar("CVrmse", sqrt(mean((st_matrix("r(est)")):^2)))
di as text   "Avg rmse     : " CVrmse

* Model 12 // non-agrarian countries
* low reliance on agriculture
xtreg nmigrq_exbalk_ln nmigrq_exbalk_ln_* $SPEI_SQ_Y0Y2 ///
  i.year i.quarter_int  if migr100_exbalk==1 & contN!=1, ///
  fe vce(cluster cowc)
est sto tabA35_m12
* F test spei parameters & aic
test L0.wmean_speiy L4.wmean_speiy L8.wmean_speiy ///
 cL0.wmean_speiy#cL0.wmean_speiy cL4.wmean_speiy#cL4.wmean_speiy cL8.wmean_speiy#cL8.wmean_speiy
estat ic
* cross validation / RMSE
local reg_call=e(cmdline)
qui crossfold `reg_call'
mata : st_numscalar("CVrmse", sqrt(mean((st_matrix("r(est)")):^2)))
di as text   "Avg rmse     : " CVrmse



*******		 Table A.36



* Model 13 // agrarian countries
xtreg nmigrq_exbalk_ln nmigrq_exbalk_ln_* spei_drought spei_hrain ///
	  i.year i.quarter_int if migr100_exbalk==1 & contN==1, fe vce(cluster cowc)
est sto tabA36_m13
* F test spei parameters & aic
test spei_drought spei_hrain
estat ic
* cross validation / RMSE
local reg_call=e(cmdline)
qui crossfold `reg_call'
mata : st_numscalar("CVrmse", sqrt(mean((st_matrix("r(est)")):^2)))
di as text   "Avg rmse     : " CVrmse

* Model 14 // non-agrarian countries
xtreg nmigrq_exbalk_ln nmigrq_exbalk_ln_* spei_drought spei_hrain  ///
  i.year i.quarter_int if migr100_exbalk==1 & contN!=1, fe vce(cluster cowc)
est sto tabA36_m14
* F test spei parameters & aic
test spei_drought spei_hrain
estat ic
* cross validation / RMSE
local reg_call=e(cmdline)
qui crossfold `reg_call'
mata : st_numscalar("CVrmse", sqrt(mean((st_matrix("r(est)")):^2)))
di as text   "Avg rmse     : " CVrmse


* Model 15 // agrarian countries
xtreg nmigrq_exbalk_ln nmigrq_exbalk_ln_* L(0 4 8).spei_drought L(0 4 8).spei_hrain ///
	  i.year i.quarter_int if migr100_exbalk==1 & contN==1, fe vce(cluster cowc)
est sto tabA36_m15
* F test spei parameters & aic
test L0.spei_drought L4.spei_drought L8.spei_drought L0.spei_hrain L4.spei_hrain L8.spei_hrain
estat ic
* cross validation / RMSE
local reg_call=e(cmdline)
qui crossfold `reg_call'
mata : st_numscalar("CVrmse", sqrt(mean((st_matrix("r(est)")):^2)))
di as text   "Avg rmse     : " CVrmse

* Model 16 // non-agrarian countries
xtreg nmigrq_exbalk_ln nmigrq_exbalk_ln_* L(0 4 8).spei_drought L(0 4 8).spei_hrain  ///
  i.year i.quarter_int if migr100_exbalk==1 & contN!=1, fe vce(cluster cowc)
est sto tabA36_m16
* F test spei parameters & aic
test L0.spei_drought L4.spei_drought L8.spei_drought L0.spei_hrain L4.spei_hrain L8.spei_hrain
estat ic
* cross validation / RMSE
local reg_call=e(cmdline)
qui crossfold `reg_call'
mata : st_numscalar("CVrmse", sqrt(mean((st_matrix("r(est)")):^2)))
di as text   "Avg rmse     : " CVrmse



*******		 Relative risk figures




* Figure A.26 (Model 7 & 8, Table A.35)

qui rr_est_Sp tabA35_m7 tabA35_m8 "quadratic"

